


Web タイルセットを使用した画像の自動登録

TNTmips のジオリファレンス処理には自動登録 (Auto-Register) の機能があり、1つの画像に対して同じ領域の参照画像を使って数百から数千のコントロールポイントを自動的に生成できます。(テクニカルガイド「ジオリファレンス：参照画像への自動登録 (Georeference: Auto-Register to Reference Image)」を参照)。これと同様の機能が TNT 地理空間スクリプト (SML) を使ったスクリプトを利用して実現できます。マイクロイメージ社はこの機能を実演する 2 種類のサンプルスクリプトを提供しています。1つは参照画像としてマイクロイメージ社のサイトにある Web タイルセットを使うもの (次ページに抜粋を掲載)、もう1つは参照画像としてローカルの TNT プロジェクトファイル中のカラー合成ラスタを使うものです。

自動登録の設定と操作は IMAGE_PIPELINE_AUTOREGISTER クラスにカプセル化されています。このクラスのメンバを用いてスクリプト作成者は「初期精度」や「最大残差」などの自動登録に必要なパラメタの全てを設定できます。サンプルスクリプト中のクラスの使用法を示す概略図を下記に示しています。マルチバンドの入力画像から参照画像と照合するバンド1つが選択され (図の右側)、画像用パイプラインソースを作るために使われます。参照画像に対してもパイプラインソースが作られます。Web タイルセットを参照画像として使用する場合、インターネット上の Web タイルセットの URL 文字列 (ローカルの Web タイルセットの場合はファイルパス) を使用してタイルセット用パイプラインソースが作られます。照合に使用するカラー成分を選別するのに選択フィルタが使われます。ローカルのコンポジット画像を参照用に使う場合は、照合に使用するカラーはパイプラインソース用のコンストラクタ内で直接指定できますので、選択フィルタは必要ありません。その結果生じた入力画像と参照画像用のパイプラインステージはパラメタとして AUTOREGISTER クラスの Run() メソッドへ送られます。それは処理を実行し、メモリ内にコントロールポイントジオリファレンスを生成します。その後 Save() メソッドを使用してこのジオリファレンスサブオブジェクトを入力画像の各バンドに書き出します。



セルサイズ 20m の SPOT4 画像 (カラー赤外線バンドを使用)。サンプル SML の自動登録処理でコントロールポイントを生成。参照画像はマイクロイメージ社サイトにある 2010 年サウスダコタ州 GM_BM Web タイルセットです (全米農業イメージプログラム (NAIP) のナチュラルカラー正射画像の郡データ)。


自動登録の設定：
 入力画像と参照画像中の緑のスペクトル成分を照合
 初期精度評価 = 20 セル
 生成するポイントの間隔 = 75 セル
 最大ポイント残差 = 1.5 セル
 相関領域サイズ = 128 セル
 最良適合モデル = 2 次多項式

自動登録処理で 537 個のコントロールポイントを生成

二乗平均平方根 (RMS) 残差：
 $X = 0.36$ セル、 $Y = 0.43$ セル、 $XY = 0.56$ セル
 (モデル = 平面投影)

平均絶対残差：
 $X = 0.28$ セル、 $Y = 0.32$ セル

自動登録サンプルスクリプトの処理概要



SML での自動登録処理はイメージパイプライン構造体を使用しますが、完全なパイプライン操作ではありません。新しい "ターゲット" 画像は作られず、リファレンス情報だけが作られます。

これらのサンプルスクリプトは TNT プロジェクトファイルにインポートされた 4 バンドのスポット衛星画像の自動登録処理用に設計されています。

しかし、これらのスクリプトは GeoTIFF や GeoJP2 のようなファイル形式の単一バンドやマルチバンド、カラーコンポジット画像を処理できるよう簡単に修正できます。また多数の入力画像を使ってバッチ処理を行うこともできます。

また、これらのスクリプトの自動登録設定は、面積やセルサイズ、形状が異なる入力画像に対しても適応することができます。ローカルの画像を参照する例では参照画像から読み込んだ座標参照系 (CRS) が自動登録処理と出力のジオリファレンスに用いられます。しかし、入力画像の CRS (指定がない場合のデフォルト) や他の CRS を使うことも可能です。自動登録処理をより一般的な画像処理スクリプトに統合し、出力画像の CRS ヘリサンプリングしたり、画像強調やその他の大量処理も行うことができます。

(次ページに続く)

www.microimages.com/downloads/scripts.htm にはダウンロード可能な多くのサンプルスクリプトがあり、スクリプトやクエリーで TNT 製品のスクリーン言語の特徴をどのように利用したらよいか解説しています。

AutoRegisterSPOT4bandWebTilesetRef.sml

(自動登録の参照画像としてインターネット上の Web タイルセットを使用しています)

```
numeric err;
```

エラーチェック処理

```
proc ReportError(numeric linenum, numeric err) {  
  printf("FAILED -line: %d, error: %d\n", linenum - 1, err);  
  PopupError(err);  
}  
clear();
```

コンソールをクリア

自動登録する画像バンドを選択

```
class STRING prompt$;  
prompt$ = "Choose 4 SPOT MS bands:";  
class RVC_OBJITEM inpObjItemList[];  
numeric numBands;
```

選択したラスタオブジェクト用の RVC_OBJITEM のハッシュ

```
DlgGetObjects(prompt$, "Raster", inpObjItemList, "ExistingOnly");
```

```
numeric i;  
for i = 1 to inpObjItemList.GetNumItems()  
  {  
    printf("Band %d = %s\n", i, inpObjItemList[i].GetObjectPath());  
  }  
}
```

ObjItem ハッシュをチェック

参照画像に使う Web タイルセットの URL を得る

```
class STRING url$, default$;  
default$ = "http://www.microimages.com/geodata/kappa/SD2010 GM state/  
SD2010 GM state.tsd";  
prompt$ = "Enter or paste URL of web tileset reference:";  
url$ = PopupString(prompt$, default$, "Web Tileset URL");
```

ステータスダイアログとそのコンテキストを作成

```
class STATUSDIALOG statusD;  
statusD.Create();  
class STATUSCONTEXT statusC;  
statusC = statusD.CreateContext();
```

プロセス状態を表示するためのステータスダイアログ

ステータスダイアログ用コンテキスト

自動登録に使用する入力画像の単一バンド用のパイプラインソースを作成。ObjItemList の一次元インデックスによって 3 = Band、3 = green を指定

```
class IMAGE_PIPELINE_SOURCE_RVC inpSource(inpObjItemList[3]);  
err = inpSource.Initialize();  
if (err < 0) then ReportError(_context.CurrentLineNum, err);  
else print("input source initialized");
```

参照 Web タイルセット画像 (ナチュラルカラー合成) 用のパイプラインソースを作成

```
class IMAGE_PIPELINE_SOURCE_TILESET refTileset(url$);  
err = refTileset.Initialize();  
if (err < 0) then ReportError(_context.CurrentLineNum, err);  
else print("reference source initialized");
```

参照タイルセットの緑のバンドを識別するフィルタ

```
array numeric srcSamples[1];  
srcSamples[1] = 2;
```

緑のバンドへのインデックス

```
class IMAGE_PIPELINE_FILTER_SELECT refTilesetGreen(refTileset,  
  srcSamples, 1);  
err = refTilesetGreen.Initialize();  
if (err < 0) then ReportError(_context.CurrentLineNum, err);  
else print("select filter for reference tileset initialized");
```

自動登録クラスのパラメータを設定。座標参照系は入力画像から自動的に設定されます

```
print("Setting up auto-register...");  
statusC.Message = "Setting up auto-register...";  
class IMAGE_PIPELINE_AUTOREGISTER autoReg;
```

```
autoReg.SetAutoGenerateGCPs(1);
```

コントロールポイントの自動生成の設定

```
autoReg.SetModel("PlaneProjective");
```

初期モデルの設定

```
autoReg.SetUseExistingGCPs(0, 0);
```

既存のコントロールポイントは使用しない (処理終了時に削除)

```
autoReg.SetInitialAccuracy(20);
```

初期精度 = 20 セル

```
autoReg.SetGCPSpacing(100);
```

ポイントの間隔 = 100 セル

```
autoReg.SetMaxGCPResidual(2);
```

最大残差 = 2 セル

```
autoReg.SetCorrPatchSize(128);
```

相関領域サイズ = 128 セル

```
autoReg.SetMaxModel("PolynomialOrder2");
```

最適適合モデルを設定

自動登録を実行

```
print("Parameters set, starting auto-register");  
statusC.Message = "Running auto-register...";  
autoReg.Run(inpSource, refSource);
```

自動登録クラスより 2 乗平均平方根 (RMS)XY 残差と結果モデルを得る

```
numeric resultRMS;  
class STRING resultModel;  
resultRMS = autoReg.GetResultRMSResidual();  
resultModel = autoReg.GetResultModel();  
printf("Result residual = %.2f\n", resultRMS);  
printf("Result model = %s\n", resultModel);
```

コントロールポイント全体の 2 乗平均平方根をチェックし、処理が終われば全ての入力のジオリファレンスを保存する

```
if (resultRMS <= 1.5)  
  {  
    print("residual less than cut-off, saving result");  
  
    for i = 1 to inpObjItemList.GetNumItems()  
      {  
        autoReg.SaveGeoreference(inpObjItemList[i]);  
        printf("saved georeference for input %d\n", i);  
      }  
  }
```

最後のステータスダイアログメッセージを設定し、ダイアログを閉じる

```
statusC.Message = "Script Ran to Completion";  
statusD.Destroy();
```

```
print("Done");
```

AutoRegisterSPOT4band.sml ではプロジェクトファイル内のローカルのカラー合成画像を参照画像として使用しています。